



Mis Cuadros de Mando (I)

Basado en Flores Cortados.

Jose Ignacio González Gómez
Departamento de Economía Financiera y Contabilidad - Universidad de La Laguna

www.jggomez.eu

INDICE

1	Creación de gráficos dinámicos basados en rangos dinámicos. “Partes de Trabajo de Flores y Plantas 1”	2
1.1	<i>Presentación del caso, ficheros: flores cortadas 1 y 2.xlsm</i>	2
1.1.1	Base de datos de partida.....	2
1.1.2	Creación de nuestra tabla dinámica.....	2
1.1.3	Creación de un rango dinámico basado en tablas dinámicas, rango trabajadores	3
1.1.4	Aplicando el rango dinámico a las celdas	4
1.1.5	Creación de un segundo rango dinámico “Flores_Cortadas”	5
1.1.6	Trabajando con el rango dinámico. Combinando con la función Coincidir	6
1.2	<i>Solución, creación del grafico dinámico</i>	6
1.2.1	Generación manual del gráfico.....	6
1.2.2	Avanzando definiendo el rango de gráfico.....	8
1.2.3	Evitando introducción de valores erróneos	11
2	Importación de datos dinámicos basado en dos o más rangos dinámicos.....	14
2.1	<i>Presentación del caso, fichero flores cortadas 3.xls</i>	14
2.1.1	Introducción.....	14
2.1.2	Creación de un rango dinámico basado en filas o cabeceras de tabla dinámica	14
2.2	<i>Objetivo</i>	15
2.3	<i>Solución. Empleo de la función importardatosdinamicos</i>	15
2.3.1	Consideraciones previas. Análisis detallado de la tabla dinámica, campos que la componen	15
2.3.2	Definir y parametrizar la función importar datos dinámicos	16
2.3.3	Caso especial total	17
3	Bibliografía, ficheros y enlaces	18

1 Creación de gráficos dinámicos basados en rangos dinámicos. “Partes de Trabajo de Flores y Plantas 1”.

1.1 Presentación del caso, ficheros: [flores cortadas 1 y 2.xlsm](#)

1.1.1 Base de datos de partida

Contamos con una base de datos que recoge los partes de trabajo desarrollado en una finca de flores y plantas. En un libro Excel recogemos a través de una consulta generada en la citada base de datos, los registros necesarios para elaborar nuestros cuadros de mando orientado a la gestión de las actividades desarrolladas en la citada empresa.

Así tal y como se muestra en la Ilustración 1 en la pestaña “BD” contamos con los registros comentados anteriormente importados de la base de datos de partes de trabajo y que hemos denominado a esta tabla con el nombre “BD_Corte_Flores”.

Id del parte	Flor o Planta	Cantidad	Trabajador Nombre	Año	Trimestre	Mes	Mes tres letras	Quincena	Sen
9	Abracadabra	1300	Ana Maria / 35	2009	4	octubre	oct	2 Quincena	44
9	Better Time	509	Ana Maria / 35	2009	4	octubre	oct	2 Quincena	44
9	Bordeux Tierra	121	Ana Maria / 35	2009	4	octubre	oct	2 Quincena	44
9	Chantelle	693	Ana Maria / 35	2009	4	octubre	oct	2 Quincena	44
9	Cherry Lip	92	Ana Maria / 35	2009	4	octubre	oct	2 Quincena	44
9	Inocencia	340	Ana Maria / 35	2009	4	octubre	oct	2 Quincena	44

Ilustración 1

Esta será la base que tomaremos como referencia para elaborar nuestras primeras tablas dinámicas así como cuadros de mando.

1.1.2 Creación de nuestra tabla dinámica

En la pestaña TD1 hemos configurado nuestra primera Tabla Dinámica que hemos llamado “TD_Cortes” tal y como se muestra en la Ilustración 2.

Flores Cortadas	Sen	43	44	45	46	47	48	49	Total
Ana Maria / 35				3.875	8.775	5.105	6.083	4.481	28.319
Anna / 10		699	12.763	11.768	8.422	9.239	6.770	2.084	51.745
Antonia / 1				3.170	2.632	3.635	5.187		14.624
Arsenia / 2				2.307	3.508	3.161	5.056	2.024	16.056
Begoña / 21				1.381	2.757		4.401	2.012	10.551
Consolación / 17				5.344	9.056	8.225	3.506	2.261	28.392
Cornelia / 14		162	928	280		1.903	2.235	415	5.923
Danuta / 22		4.184	24.814	24.151	15.715	8.646	8.714	2.942	89.166
Fatima / 43		1.008	1.715						2.723
Francisca / 36				1.959	4.024	2.799	3.386	5.245	19.444
Francisco / 28						1.251			1.251
Mateo / 99						220			220
Meme Mª Remedios / 34		2.565	19.794	16.475	14.616	10.839	5.543	2.050	71.882
Narcisa Jesus / 20				8.992	10.845	6.275	6.195		32.307
Paffience / 15		695	1.242						1.937
Silvestre / 41				2.256		1.312	1.127		4.695

Ilustración 2

En ella aparecen todos los trabajadores y las cantidades cortadas de rosas en cada semana. Como se pueden observar hemos quitado el total para las columnas.

1.1.3 Creación de un rango dinámico basado en tablas dinámicas, rango trabajadores

Una de las utilidades más empleadas a la hora de trabajar en Excel son los 'rangos dinámicos'.

Sabemos que podemos nombrar los rangos que utilizamos, pero lo habitual es que estos rangos no sean fijos, es decir, no tenga siempre (a lo largo de nuestro trabajo con la Base de datos en cuestión) la misma dimensión, i.e., el mismo número de registros. Es en este punto cuando Excel nos proporciona la posibilidad de crear un rango dinámico. Y lo haremos desarrollando la función DESREF, anidando otra función importante como CONTARA en ella.

Nos interesa a continuación crear en una celda de la misma hoja la relación de trabajadores disponibles tal y como mostramos en la Ilustración 3. En este caso hemos propuesto dos soluciones.

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2		Flores Cortadas	Sen									
3		Trabajadores	43	44	45	46	47	48	49	Total		Trabajador
4		Ana Maria / 35			3.875	8.775	5.105	6.083	4.481	28.319		Ana Maria / 35
5		Anna / 10	699	12.763	11.768	8.422	9.239	6.770	2.084	51.745		
6		Antonia / 1			3.170	2.632	3.635	5.187		14.624		Ana Maria / 35
7		Arsenia / 2			2.307	3.508	3.161	5.056	2.024	16.056		Anna / 10
8		Begoña / 21			1.381	2.757		4.401	2.012	10.551		Antonia / 1
9		Consolación / 17			5.344	9.056	8.225	3.506	2.261	28.392		Arsenia / 2
10		Cornelia / 14	162	928	280		1.903	2.235	415	5.923		Begoña / 21
11												Consolación / 17
12												Cornelia / 14

Ilustración 3

- La primera propuesta consiste en limitar (validar datos) asociados a la celda para aquellos trabajadores disponibles o activados en la tabla dinámica.
- La segunda propuesta es a través de insertar un cuadro de lista asociado en una casilla.

En ambos casos la condición es que la relación de trabajadores quede limitada a la lista seleccionada en la tabla dinámica y para ello vamos a crear un rango dinámico que llamaremos "trabajadores" cuya peculiaridad será que si el número de trabajadores seleccionado en la tabla dinámica cambia, la lista de trabajadores o rango dinámico deberá ajustarse automática (comparar Ilustración 3 e Ilustración 4)

	A	B	C	D	E	F	G	H	I	J	K	L
1												
2		Flores Cortadas	Sen									
3		Trabajadores	43	44	45	46	47	48	49	Total		Trabajador
4		Consolación / 17			5.344	9.056	8.225	3.506	2.261	28.392		Ana Maria / 35
5		Danuta / 22	4.184	24.814	24.151	15.715	8.646	8.714	2.942	89.166		
6		Fatima / 43	1.008	1.715						2.723		Consolación / 17
7		Francisco / 28				1.251				1.251		Danuta / 22
8		Mateo / 99				220				220		Fatima / 43
9												Francisco / 28
10												Mateo / 99

Ilustración 4




Ilustración 5

Uso de la función Desref y Contara. Para crear este rango dinámico adaptado a los valores contenidos en una tabla dinámica hemos hecho uso de dos funciones, Desref y Contara tal y como se muestra en la Ilustración 5. Así en concreto el rango dinámico “Trabajadores” hace referencia a la siguiente función:

$$= \text{DESREF}(\text{'TD1'!\$B\$4;0;0;CONTARA(\text{'TD1'!\$B:\$B})-2;1)$$

Vamos a explicar la función creada para este rango dinámico trabajadores.

En primer lugar definimos el nombre que queremos dar a nuestro rango dinámico, tal y como se muestra en la Ilustración 6.




Ilustración 6

- **DESREF('TD1'!\$B\$4;0;0;1)** Esta función nos permite hacer referencia a un determinado rango y para ello comenzamos en primer lugar por fijar el punto de inicio que es la celda B4 correspondiente a la hoja TD1 de nuestro libro de trabajo ('TD1'!\$B\$4). Luego tenemos que indicarle cuantas columnas o filas a la derecha tiene y en nuestro caso no tiene ninguna por eso colocamos 0,0. Solo nos falta indicar cuantas filas hacia abajo tiene que considerar y como en nuestro caso es variables, es decir depende del número de trabajadores seleccionados en la tabla dinámica es necesario usar la función Contara y unirla a la función Desref.
- **CONTARA('TD1'!\$B:\$B)-2** Es decir, a continuación debemos indicarle cuantas filas hacia abajo tiene que contar la referencia y como es interactivo, es decir en la tabla dinámica puede aumentar o disminuir el número de elementos mostrado (número de trabajadores) necesitamos usar y combinar con la función Contara que aplicándolo a toda la columna nos contara el numero de celdas no vacías que esa columna (B) contiene pero como las primeras filas contiene los encabezados es necesario por tanto restar -2.
- Finalmente unimos las dos funciones y como queremos hacer referencia a una sola columna colocamos uno al final de la formula.

1.1.4 Aplicando el rango dinámico a las celdas

Como comentamos anteriormente, queremos ahora limitar el rango de las celdas L4 y L6 tal y como se muestra en la Ilustración 4.

La primera propuesta consiste en limitar los datos a insertar en la celda L4 a través de la opción validar datos (Ilustración 7).

La segunda propuesta es a través de insertar un cuadro de lista asociado en una casilla, en este caso en L6, Ilustración 8.




Ilustración 7




Ilustración 8

1.1.5 Creación de un segundo rango dinámico “Flores_Cortadas”

Bueno el ejemplo anterior lo hemos modificado sensiblemente y vamos a crear un nuevo rango dinámico, pero esta vez la tabla dinámica no presenta las semanas sino el total de cortes por trabajador, tal y como vemos en la Ilustración 9. Al desaparecer en la tabla dinámica la fila semana (Ilustración 4) las referencias de rango dinámicas tienen que ser adaptada y por tanto hemos tenido concretamente que cambiar la de Trabajador para ajustarnos a la nueva característica de la tabla dinámica.

	A	B	C	D	E
1					
2		Trabajadores	Flores Cortadas		Trabajador
3		Ana Maria / 35	28.319		Consolación / 17
4		Anna / 10	51.745		
5					




Ilustración 9

=DESREF('TD1!\$B\$3;0;0;CONTARA('TD1!\$B:\$B)-1;1)

A continuación vamos a crear un segundo rango dinámico que llamaremos “Flores_Cortadas” y procederemos de igual forma como hemos realizado para el caso de Trabajador, es decir el resultado será:



Ilustración 10

=DESREF('TD1!\$C\$3;0;0;CONTARA('TD1!\$C:\$C)-1;1)

1.1.6 Trabajando con el rango dinámico. Combinando con la función Coincidir

Tal y como se muestra en la Ilustración 11, nuestro objetivo a continuación es que seleccionado un trabajador en la celda L4 nos muestre en la celda contigua el total de flores cortadas, para ello es necesario combinar la función DERSREF con la función COINCIDIR.

	A	B	C	D	L	M
1						
2		Flores Cortadas	Sen			
3		Trabajadores	48	Total	Trabajador	
4		Ana Maria / 35	4.481	4.481	Antonia / 1	5187
5		Anna / 10	6.770	6.770	=DESREF(TD1!\$B\$4;COINCIDIR(L4;Trabajador;2)-1;1;1)	
6		Antonia / 1	5.187	5.187	Anna / 10	

Ilustración 11

Por tanto en la celda M4 pondremos la siguiente fórmula:

=DESREF('TD1!\$B\$4;COINCIDIR(L4;Trabajador;0)-1;1;1)

Esta función viene a decir:

- Anclada la búsqueda en la celda correspondiente al primer trabajador de la tabla dinámica **'TD1!\$B\$4**
- Moverse hasta el trabajador que coincide con el seleccionado en la celda L4 sin desplazarse hacia abajo. **COINCIDIR(L4;Trabajador;0)-1**
- Y mostrar el valor de la columna siguiente **1**

1.2 Solución, creación del grafico dinámico

1.2.1 Generación manual del gráfico

Vamos a crear a continuación un gráfico simple manualmente, en primer lugar lo creamos en blanco para después modificarlo a nuestro antojo, tal y como se muestra en la Ilustración 12..

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2		Trabajadores	Flores Cortadas		Trabajador								
3		Ana Maria / 35	28.319		Consolación / 17								
4		Anna / 10	51.745										
5													
6													
7													
8													
9													
10													
11													
12													
13													
14													
15													
16													

Ilustración 12

A continuación vamos adaptarlo a nuestros rangos dinámicos, siguiendo lo mostrado en las siguientes ilustraciones.




Ilustración 13




Ilustración 14

Y de forma automática nos habrá vinculado la Serie 1 con el rango dinámico Flores_Cortadas, tal y como se muestra Ilustración 15 donde asigna el nombre del libro Excel y además el rango dinámico de la serie Flores_Cortadas.




Ilustración 15

Finalmente para personalizar debemos asignar el rango dinámico "Trabajadores" a los rótulos del eje, para ello pulsamos sobre categoría para personalizarla. Una vez más Excel corrige la fórmula propuesta añadiendo el nombre del fichero Excel de referencia (Ilustración 16).




Ilustración 16




Ilustración 17

El resultado final será un gráfico dinámico basado en rangos dinámicos.




Ilustración 18



Ilustración 19

1.2.2 Avanzando definiendo el rango de gráfico.

A continuación vamos a mejorar gráfico dinámico anterior e incorporaremos dos variables “Desde el Trabajador” – “Hasta el Trabajador” para que el grafico muestre los rangos de los trabajadores deseados, es decir queremos indicarle al gráfico que se inicie el rango de datos a mostrar con el trabajador 1 y extienda el rango a los 5 trabajadores siguientes tal y como establece el limite trabajador 2 (Ilustración 20).




Ilustración 20

Por tanto vamos a asignar un nombre a estas dos variables que serán trabajador1 y trabajador2 respectivamente. Para ello vamos a la opción correspondiente del programa y

asignamos el nombre referenciamos a la celda correspondiente tal y como muestra la Ilustración 21. Ahora tenemos que integrar estos nuevos rangos a la función que nos genera el gráfico.




Ilustración 21

Antes de continuar hemos optado por copiar el gráfico anterior para trabajar con la copia de esta forma tenemos dos gráficos uno con trazo azul (original) y otro con trazo rojo que es la copia del anterior.




Ilustración 22

Es decir, los siguientes cambios se realizan sobre el gráfico trazado en rojo con el fin de preservar los datos originales del gráfico de área azul.




Ilustración 23

Como podemos observar según el Administrador de nombres, en la Ilustración 23 estos son los rangos definidos, en este caso y para trabajar con el grafico rojo vamos a crear una copia de los dos rango dinámicos Flores_Cortadas y Trabajador y los llamaremos Cflores y Ctrabajador, conservando las formulas evidentemente.




Ilustración 24

Por tanto el grafico azul hace referencia al rango dinámico Flores_Cortadas y el grafico rojo hace referencia al rango dinámico Cflores, igualmente pasa con el rango Trabajador y Ctrabajador.




Ilustración 25

Ahora nos centramos en el gráfico rojo, queremos que las variables trabajador1 y trabajador2 definidas anteriormente actúen sobre la serie, para ello es necesario modificar los rangos dinámicos e incorporar las limitaciones impuestas.

Así partimos de Cflores cuya fórmula original, igual que la de Flores_Cortadas es:

$$\text{DESREF}(\text{TD1}!\$C\$3;0;0;\text{CONTARA}(\text{TD1}!\$C:\$C)-1;1)$$

Los valores en rojo serán los que sustituimos de tal forma que la función modificada quedará, para el caso Cflores:

$$=\text{DESREF}(\text{TD1}!\$C\$3;\text{trabajador1};0;\text{trabajador2}-\text{trabajador1}+1;1)$$

Es decir la primera condición (valor 0) por trabajador1, por tanto lo que estamos diciendo es que se coloque en el valor del trabajador1 seleccionado y a partir de ahí comienza el rango de datos.

También la condición limitada por contara será sustituida, ahora nos interesa que nos dibuje un rango de valores que se corresponde a la diferencia entre el valor asignado al trabajador2 menos el valor asignado al trabajador1 y como queremos incorporar a los extremos sumamos 1, por eso la formula queda como trabajador2-trabajador1+1.

De igual forma procedemos a modificar el rango Ctrabajador, mostramos a continuación los cambios respectivos en rojo lo que debemos y hemos modificado respectivamente:

Original: $=\text{DESREF}(\text{TD1}!\$B\$3;0;0;\text{CONTARA}(\text{TD1}!\$B:\$B)-1;1)$

Adaptada: $=\text{DESREF}(\text{TD1}!\$B\$3;\text{trabajador1};0;\text{trabajador2}-\text{trabajador1}+1;1)$

1.2.3 Evitando introducción de valores erróneos

Si por un casual, el usuario no introduce valores correctos, esto nos daría un gráfico en blanco ya que el rango no se podría dibujar. Tal y como vemos en la Ilustración 26, contamos en la tabla dinámica con solo 4 registros y le estamos indicando que nos dibuje desde el trabajador 4 hasta el 7 ambos inclusive, evidentemente carece de datos para realizar el gráfico.




Ilustración 26

Por tanto una de las condiciones será que el valor en la casilla E7 no puede ser mayor que el número de registros.

Para hacer esto hemos creado dos variables (Ver Ilustración 27):

- Nº de registros – celda G6: Cuenta el numero de registros que tiene la tabla dinámica, para ello insertamos en la celda G6 la función: =CONTARA('TD1'!\$B:\$B)-1 que nos dará el número total de registros que se muestran en la tabla dinámica de referencia .
- Valor máximo – celda G9: El valor a representar “Hasta el Trabajador” no podrá ser mayor que la diferencia entre =Nº de registros-trabajador1, así la formula en la celda G9 es = G6-trabajador1




Ilustración 27

2 Importación de datos dinámicos basado en dos o más rangos dinámicos

2.1 Presentación del caso, fichero *flores cortadas 3.xls*

2.1.1 Introducción

En el presenta caso vamos a recuperar la Ilustración 12, correspondiente al libro Excel Flores cortadas 1 y lo pasamos a llamar Flores cortadas 3 que es sobre el que vamos a trabajar a continuación.

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2		Flores Cortadas	Sen										
3		Trabajadores	43	44	45	46	47	48	49	Total			
4		Anna / 10	699	12.763	11.768	8.422	9.239	6.770	2.084	51.745		Trabajador	
5		Antonia / 1			3.170	2.632	3.635	5.187		14.624		Danuta / 22	
6		Consolación / 17			5.344	9.056	8.225	3.506	2.261	28.392		Anna / 10	
7		Cornelia / 14	162	928	280		1.903	2.235	415	5.923		Antonia / 1	
8		Danuta / 22	4.184	24.814	24.151	15.715	8.646	8.714	2.942	89.166		Consolación / 17	
9												Danuta / 22	

Ilustración 28

Como podemos observar en la Ilustración 28 hemos realizado una pequeña modificación ya que en la tabla dinámica hemos incorporado la variable Semana del Año.

2.1.2 Creación de un rango dinámico basado en filas o cabeceras de tabla dinámica

Por tanto, en este caso nos interesa crear un rango dinámico asociado a la citada tabla dinámica similar al creado anteriormente para Trabajador y que llamaremos Periodo y que contendrá el rango dinámico correspondiente a las semanas del año filtradas en la tabla dinámica, para ello vamos administrador de nombres tal y como hemos realizado para el caso de Trabajadores y la función a incorporar en este caso será:




Ilustración 29

Es decir la formula a emplear será:

$$=DESREF('TD1'!\$C\$3;;;CONTARA('TD1'!\$3:\$3)-1)$$

De esta forma hemos creado dos rangos dinámicos asociados a la citada tabla de forma que si se amplían o disminuyen los filtros tanto de trabajadores como de semanas del año estos rangos al ser dinámicos asociados a la citada tabla se actualizarán de forma automática.

Trabajadores	43	44	45	46	47	48	49	Total
Anna / 10	699	12.763	11.768	8.422	9.239	6.770	2.084	51.745
Antonia / 1			3.170	2.632	3.635	5.187		14.624
Consolación / 17			5.344	9.056	8.225	3.506	2.261	28.392
Cornelia / 14	162	928	280		1.903	2.235	415	5.923
Danuta / 22	4.184	24.814	24.151	15.715	8.646	8.714	2.942	89.166

Ilustración 30

Así con estos dos rangos dinámicos el usuario podrá seleccionar tanto el trabajador como la semana concreta, adaptado lógicamente a los valores disponibles del filtro establecido en la propia tabla dinámica.

2.2 Objetivo

Ya que contamos con dos filtros, nos interesa que una vez seleccionado el trabajador y la semana, poder usar una función que nos de el total de cortes de flores considerando esas dos condiciones.

Es decir retomando la Ilustración 30 deseamos que en la celda N5 (con relleno gris) nos determinara el valor correspondiente a los cortes de flores correspondiente al trabajador Danuta/22 en la semana 45 que ha seleccionado el usuario y que por tanto nos tendría que devolver el valor 24.151.

2.3 Solución. Empleo de la función importardatosdinamicos

2.3.1 Consideraciones previas. Análisis detallado de la tabla dinámica, campos que la componen

Para alcanzar nuestro objetivo contamos con una función específica para importar datos de una tabla dinámica, esta función es importardatosdinamicos o conocida también como getpivotdata.

Pero antes de explicar el uso de la citada función vamos a exponer brevemente las bases de la tabla dinámica, en concreto los campos disponibles y el diseño de la misma.

Para acceder a esta información basta con situarnos dentro del cualquier punto de nuestra TD y seleccionar la pestaña Opciones y después el botón Lista de Campos que nos dará acceso a la información básica tal y como se muestra en la




Ilustración 31.

Así para el caso que nos ocupa, debemos tener en cuenta que nuestra tabla se basa en los campos Cantidad, Trabajador Nombre y Semana del Año

Es necesario tener presente estos datos para una correcta parametrización de la función importar datos dinámicos como analizaremos a continuación.




Ilustración 31

2.3.2 Definir y parametrizar la función importar datos dinámicos

Como hemos dicho anteriormente, la función a aplicar para alcanzar nuestro objetivo es **IMPORTARDATOSDINAMICOS**, esta función nos permitirá traer los datos específicos que nos interesan (en nuestro caso la cantidad de rosas cortadas), según los criterios de selección establecidos en las celdas L5 y L7 correspondientes al trabajador y la semana (ver Ilustración 30).

Esta función exige como mínimo dos parámetros obligatorios que son, el primero el nombre del campo que contiene los datos en la tabla dinámica (en nuestro caso Cantidad, ver configuración de la TD en la Ilustración 31) y el segundo una referencia a una celda de la tabla dinámica (hemos escogido la celda B8, pero podríamos haber seleccionado cualquier otra dentro de la td), esto es necesario para saber que tabla dinámica se va a utilizar.

Luego se pueden agregar hasta 126 pares de parámetros donde se indica el nombre de un campo y un valor para filtrar los datos. En nuestro caso hemos seleccionado dos campos, tal y como se muestra en la Ilustración 32 y que son:

- Campo Filtro “Trabajador Nombre” que así se define internamente en la TD (ver Ilustración 31) y el elemento a filtrar lo tomara de la celda L5 que se corresponde con el rango dinámico trabajadores seleccionado por el usuario.
- Campo Filtro “Semana del Año” que así se define también internamente en la TD (ver Ilustración 31) y el elemento a filtrar lo tomara de la celda L7 que se corresponde con el rango dinámico periodos seleccionado por el usuario.




Ilustración 32

El resultado de ello es la siguiente función adaptada a nuestro caso (celda N5):

=IMPORTARDATOSDINAMICOS("Cantidad";\$B\$4;"Trabajador Nombre";L5;"Semana del Año";L7)

Que viene a significar en términos textuales algo similar a:

"... extrae el valor cantidad de la tabla dinámica correspondiente a la celda b4 donde el criterio "Trabajador Nombre" es el contenido en la celda L5 y el criterio "Semana del Año" se corresponde con el valor de la celda L7"

Trabajador	Total Cortes	Trabajador	Total Cortes	Trabajador	Total Cortes
Cornelia / 14	2.235	Anna / 10	2.084	Danuta / 22	4.184
Semana		Semana		Semana	
48		49		43	

Ilustración 33

En la Ilustración 33 presentamos tres casos sobre cómo queda integrada la función en nuestra celda objetivo (N5) y que puede contrastar el resultado con los valores de la Ilustración 30.

2.3.3 Caso especial total

Todo funcionará perfectamente excepto en el caso de seleccionar Total en el criterio "Semana del Año" por cuanto que la función no entiende el valor total arrojando un error.

Trabajador	Total Cortes
Danuta / 22	#¡REF!
Semana	
Total	

Ilustración 34

En este caso especial que queremos determinar el total de cortes de rosas para un trabajador concreto, la adaptación de la función será:

=IMPORTARDATOSDINAMICOS("Cantidad";\$B\$4;"Trabajador Nombre";L5)

Esto significa que debemos establecer una condición en la celda correspondiente a Total Cortes (N5):

- SI Semana (celda L7) es Total entonces
 - IMPORTARDATOSDINAMICOS("Cantidad";\$B\$4;"Trabajador Nombre";L5)
- En caso contrario
 - =IMPORTARDATOSDINAMICOS("Cantidad";\$B\$4;"Trabajador Nombre";L5;"Semana del Año";L7)

Por tanto, para solventar este problema el valor final de la celda N5 será:

=SI(L7="Total";IMPORTARDATOSDINAMICOS("Cantidad";\$B\$4;"Trabajador Nombre";L5);IMPORTARDATOSDINAMICOS("Cantidad";\$B\$4;"Trabajador Nombre";L5;"Semana del Año";L7))

Trabajador	Total Cortes	Trabajador	Total Cortes
Danuta / 22	15.715	Danuta / 22	89.166
Semana		Semana	
46		Total	

Ilustración 35

3 Bibliografía, ficheros y enlaces

Bibliografía

Ficheros

- Flores Cortadas 1.xlsm
- Flores Cortadas 2.xlsm
- Flores Cortadas 3.xlsm
-

Enlaces

<http://economia-excel.blogspot.com/2010/01/graficos-de-ventas.html>

<http://jldexcelsp.blogspot.com/2006/09/la-funcin-desref-offset-de-excel-una.html>

<http://lareboticadeexcel.blogspot.com/2009/04/calculos-con-rangos-dinamicos-desref-y.html>